16

The Linear Tetrahedron

The Linear (4-Node) Tetrahedron



Volume of tetrahedron is

$$\mathcal{V} = \frac{1}{6} \det \begin{bmatrix} 1 & 1 & 1 & 1 \\ x_1 & x_2 & x_3 & x_4 \\ y_1 & y_2 & y_3 & y_4 \\ z_1 & z_2 & z_3 & z_4 \end{bmatrix}$$

This is a signed quantity and must be >0

Tetrahedron Natural Coordinates



interpretation as a moving surface



interpretation as volume ratio

Tetrahedron Coordinates (cont'd)

Linear interpolation

$$F\,(\zeta_1,\ \zeta_2,\ \zeta_3,\ \zeta_4) = F_1\zeta_1 + F_2\zeta_2 + F_3\zeta_3 + F_4\zeta_4 = F_i\,\zeta_i$$

Make $F = 1,x,y,z$ above and collect as matrix relation:

$$\begin{bmatrix} 1 \\ x \\ y \\ z \end{bmatrix} = \begin{bmatrix} 1 & 1 & 1 & 1 \\ x_1 & x_2 & x_3 & x_4 \\ y_1 & y_2 & y_3 & y_4 \\ z_1 & z_2 & z_3 & z_4 \end{bmatrix} \begin{bmatrix} \zeta_1 \\ \zeta_2 \\ \zeta_3 \\ \zeta_4 \end{bmatrix}$$

Inverting

$$\begin{bmatrix} \zeta_1 \\ \zeta_2 \\ \zeta_3 \\ \zeta_4 \end{bmatrix} = \frac{1}{6\mathcal{V}} \begin{bmatrix} 6\mathcal{V}_1 & a_1 & b_1 & c_1 \\ 6\mathcal{V}_2 & a_2 & b_2 & c_2 \\ 6\mathcal{V}_3 & a_3 & b_3 & c_3 \\ 6\mathcal{V}_4 & a_4 & b_4 & c_4 \end{bmatrix} \begin{bmatrix} 1 \\ x \\ y \\ z \end{bmatrix}$$

Partial Derivatives

$$\frac{\partial x}{\partial \zeta_i} = x_i, \quad \frac{\partial y}{\partial \zeta_i} = y_i, \quad \frac{\partial z}{\partial \zeta_i} = z_i$$

$$6\mathcal{V}\frac{\partial \zeta_i}{\partial x} = a_i, \quad 6\mathcal{V}\frac{\partial \zeta_i}{\partial y} = b_i, \quad 6\mathcal{V}\frac{\partial \zeta_i}{\partial z} = c_i$$

For any differentiable function *F(x,y,z)* given over the element, use chain rule to get the Cartesian partials:

$$\frac{\partial F}{\partial x} = \frac{1}{6\mathcal{V}}\left(\frac{\partial F}{\partial \zeta_1}a_1 + \frac{\partial F}{\partial \zeta_2}a_2 + \frac{\partial F}{\partial \zeta_3}a_3 + \frac{\partial F}{\partial \zeta_4}a_4\right) = \frac{1}{6\mathcal{V}}\frac{\partial F}{\partial \zeta_i}a_i$$

$$\frac{\partial F}{\partial y} = \frac{1}{6\mathcal{V}}\left(\frac{\partial F}{\partial \zeta_1}b_1 + \frac{\partial F}{\partial \zeta_2}b_2 + \frac{\partial F}{\partial \zeta_3}b_3 + \frac{\partial F}{\partial \zeta_4}b_4\right) = \frac{1}{6\mathcal{V}}\frac{\partial F}{\partial \zeta_i}b_i$$

$$\frac{\partial F}{\partial z} = \frac{1}{6\mathcal{V}}\left(\frac{\partial F}{\partial \zeta_1}c_1 + \frac{\partial F}{\partial \zeta_2}c_2 + \frac{\partial F}{\partial \zeta_3}c_3 + \frac{\partial F}{\partial \zeta_4}c_4\right) = \frac{1}{6\mathcal{V}}\frac{\partial F}{\partial \zeta_i}c_i$$

Displacement Interpolation for Use in TPE Principle

$$\begin{bmatrix} u_x \\ u_y \\ u_z \end{bmatrix} = \begin{bmatrix} u_{x1} & u_{x2} & u_{x3} & u_{x4} \\ u_{y1} & u_{y2} & u_{y3} & u_{y4} \\ u_{z1} & u_{z2} & u_{z3} & u_{z4} \end{bmatrix} \begin{bmatrix} \zeta_1 \\ \zeta_2 \\ \zeta_3 \\ \zeta_4 \end{bmatrix}$$

Appending to the geometry definition one has the iso-P definition of the linear tetrahedron

$$\begin{bmatrix} 1 \\ x \\ y \\ z \\ u_x \\ u_y \\ u_z \end{bmatrix} = \begin{bmatrix} 1 & 1 & 1 & 1 \\ x_1 & x_2 & x_3 & x_4 \\ y_1 & y_2 & y_3 & y_4 \\ z_1 & z_2 & z_3 & z_4 \\ u_{x1} & u_{x2} & u_{x3} & u_{x4} \\ u_{y1} & u_{y2} & u_{y3} & u_{y4} \\ u_{z1} & u_{z2} & u_{z3} & u_{z4} \end{bmatrix} \begin{bmatrix} \zeta_1 \\ \zeta_2 \\ \zeta_3 \\ \zeta_4 \end{bmatrix}$$

Strain-Displacement Relations

$$\mathbf{e} = \begin{bmatrix} e_{xx} \\ e_{yy} \\ e_{zz} \\ 2e_{xy} \\ 2e_{yz} \\ 2e_{zx} \end{bmatrix} = \begin{bmatrix} \partial/\partial x & 0 & 0 \\ 0 & \partial/\partial y & 0 \\ 0 & 0 & \partial/\partial z \\ \partial/\partial y & \partial/\partial x & 0 \\ 0 & \partial/\partial z & \partial/\partial y \\ \partial/\partial z & 0 & \partial/\partial x \end{bmatrix} \begin{bmatrix} u_x \\ u_y \\ u_z \end{bmatrix} = \mathbf{D}\,\vec{\mathbf{u}}.$$

Inserting the displacement field interpolation one gets

$$\mathbf{e} = \mathbf{B}\,\mathbf{u}^e$$

Strain-Displacement Relations (cont'd)

If the element node displacement is arranged component-wise:

$$\mathbf{u}^e = [\,u_{x1} \quad u_{x2} \quad u_{x3} \quad u_{x4} \quad u_{y1} \quad u_{y2} \quad \cdots \quad u_{z4}\,]^T$$

$$\mathbf{B} = \frac{1}{6\mathcal{V}} \begin{bmatrix} a_1 & a_2 & a_3 & a_4 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & b_1 & b_2 & b_3 & b_4 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & c_1 & c_2 & c_3 & c_4 \\ b_1 & b_2 & b_3 & b_4 & a_1 & a_2 & a_3 & a_4 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & c_1 & c_2 & c_3 & c_4 & b_1 & b_2 & b_3 & b_4 \\ c_1 & c_2 & c_3 & c_4 & 0 & 0 & 0 & 0 & a_1 & a_2 & a_3 & a_4 \end{bmatrix}$$

If the element node displacement is arranged node-wise:

$$\mathbf{u}^e = [\,u_{x1} \quad u_{y1} \quad u_{z1} \quad u_{x2} \quad u_{y2} \quad u_{z2} \quad \cdots \quad u_{z4}\,]^T$$

$$\mathbf{B} = \frac{1}{6\mathcal{V}} \begin{bmatrix} a_1 & 0 & 0 & a_2 & 0 & 0 & a_3 & 0 & 0 & a_4 & 0 & 0 \\ 0 & b_1 & 0 & 0 & b_2 & 0 & 0 & b_3 & 0 & 0 & b_4 & 0 \\ 0 & 0 & c_1 & 0 & 0 & c_2 & 0 & 0 & c_3 & 0 & 0 & c_4 \\ b_1 & a_1 & 0 & b_2 & a_2 & 0 & b_3 & a_3 & 0 & b_4 & a_4 & 0 \\ 0 & c_1 & b_1 & 0 & c_2 & b_2 & 0 & c_3 & b_3 & 0 & c_4 & b_4 \\ c_1 & 0 & a_1 & c_2 & 0 & a_2 & c_3 & 0 & a_3 & c_4 & 0 & a_4 \end{bmatrix}$$

Stress-Strain (Constitutive) Equations

$$\sigma_{ij} = E_{ijk\ell} e_{k\ell}$$

$$\boldsymbol{\sigma} = [\, \sigma_{11} \quad \sigma_{22} \quad \sigma_{33} \quad \sigma_{12} \quad \sigma_{23} \quad \sigma_{31} \,]^T =$$
$$= [\, \sigma_{xx} \quad \sigma_{yy} \quad \sigma_{zz} \quad \sigma_{xy} \quad \sigma_{yz} \quad \sigma_{zx} \,]^T$$

$$\begin{bmatrix} \sigma_{xx} \\ \sigma_{yy} \\ \sigma_{zz} \\ \sigma_{xy} \\ \sigma_{yz} \\ \sigma_{zx} \end{bmatrix} = \begin{bmatrix} E_{11} & E_{12} & E_{13} & E_{14} & E_{15} & E_{16} \\ & E_{22} & E_{23} & E_{24} & E_{25} & E_{26} \\ & & E_{33} & E_{34} & E_{35} & E_{36} \\ & & & E_{44} & E_{45} & E_{46} \\ & & & & E_{55} & E_{56} \\ \text{symm} & & & & & E_{66} \end{bmatrix} \begin{bmatrix} e_{xx} \\ e_{yy} \\ e_{zz} \\ 2e_{xy} \\ 2e_{yz} \\ 2e_{zx} \end{bmatrix}$$

$$\boldsymbol{\sigma} = \mathbf{E}\,\mathbf{e}$$

Element Stiffness Matrix from TPE

$$\mathbf{K}^e = \int_{V^e} \mathbf{B}^T \mathbf{E} \mathbf{B} \, dV$$

If the elasticity matrix **E** is constant over element:

$$\mathbf{K}^e = \mathcal{V} \, \mathbf{B}^T \mathbf{E} \mathbf{B}$$

This stiffness matrix is 12 x 12, and has rank 6 (assuming **E** and **B** have full rank)

Element Consistent Node Force Vector for Body Force Field

$$\mathbf{b} = \begin{bmatrix} b_x \\ b_y \\ b_z \end{bmatrix} \qquad \begin{bmatrix} u_x \\ u_y \\ u_z \end{bmatrix} = \mathbf{N}\,\mathbf{u}^e$$

The TPE principle gives

$$\mathbf{f}^e = \int_{V^e} \mathbf{N}^T\,\mathbf{b}\;dV$$

Element Consistent Node Force Vector for Body Force Field (cont'd)

If the node displacement vector is arranged component-wise:

$$\mathbf{N} = \begin{bmatrix} \zeta_1 & \zeta_2 & \zeta_3 & \zeta_4 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & \zeta_1 & \zeta_2 & \zeta_3 & \zeta_4 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \zeta_1 & \zeta_2 & \zeta_3 & \zeta_4 \end{bmatrix}$$

If the node displacement vector is arranged node-wise:

$$\mathbf{N} = \begin{bmatrix} \zeta_1 & 0 & 0 & \zeta_2 & 0 & 0 & \zeta_3 & 0 & 0 & \zeta_4 & 0 & 0 \\ 0 & \zeta_1 & 0 & 0 & \zeta_2 & 0 & 0 & \zeta_3 & 0 & 0 & \zeta_4 & 0 \\ 0 & 0 & \zeta_1 & 0 & 0 & \zeta_2 & 0 & 0 & \zeta_3 & 0 & 0 & \zeta_4 \end{bmatrix}$$

Useful Exact Volume-Integration Formulas in Terms of Tetrahedron Coordinates

Linear integrand

$$\int_{V^e} \zeta_i \, dV = \mathcal{V}/4$$

Quadratic integrand

$$\int_{V^e} \zeta_i \, \zeta_j \, dV = \begin{cases} \mathcal{V}/10 & \text{if } i = j \\ \mathcal{V}/20 & \text{if } i \neq j \end{cases}$$

General formula for a monomial term

$$\int_{V^e} \zeta_1^i \, \zeta_2^j \zeta_3^k \zeta_4^\ell \, dV = \frac{i! \; j! \; k! \; \ell!}{(i + j + k + \ell + 3)!} 6\mathcal{V}$$

Tetr4 Stiffness Matrix Module

```
IsoTetr4Stiffness[ncoor_,Emat_,{},options_]:= Module[{i,n=4,nf=12,
   k,c,w,Jdet,zetalist,xyzlist,numer,Bx,By,Bz,Be,Ke},
  If [Length[options]>0, numer=options[[1]]];
  xyzlist={Table[ncoor[[i,1]],{i,n}],Table[ncoor[[i,2]],{i,n}],
           Table[ncoor[[i,3]],{i,n}]}; Ke=Table[0,{nf},{nf}];
  {Bx,By,Bz,Jdet}=IsoTetr4ShapeFunCarDer[xyzlist,{},numer];
  Be={Flatten[Table[{Bx[[i]],0,      0      },{i,n}]],
      Flatten[Table[{0,      By[[i]],0      },{i,n}]],
      Flatten[Table[{0,      0,      Bz[[i]]},{i,n}]],
      Flatten[Table[{By[[i]],Bx[[i]],0      },{i,n}]],
      Flatten[Table[{0,      Bz[[i]],By[[i]]},{i,n}]],
      Flatten[Table[{Bz[[i]],0,      Bx[[i]]},{i,n}]]};
   Ke=(Jdet/6)*Transpose[Be].(Emat.Be);
   If [!numer,Ke=Simplify[Ke]]; Return[Ke]
   ];
```

Shape Function Cartesian Derivatives Module

```
IsoTetr4ShapeFunCarDer[{xn_,yn_,zn_},zetalist_,numer_]:=
  Module[{dNz1,dNz2,dNz3,dNz4,Jmat,J11,J12,J13,J14,
   J21,J22,J23,J24,J31,J32,J33,J34,Jinv,Jdet,Bx,By,Bz},
   {dNz1,dNz2,dNz3,dNz4}={{1,0,0,0},{0,1,0,0},{0,0,1,0},{0,0,0,1}};
   J11=dNz1.xn; J12=dNz2.xn; J13=dNz3.xn;  J14=dNz4.xn;
   J21=dNz1.yn; J22=dNz2.yn; J23=dNz3.yn;  J24=dNz4.yn;
   J31=dNz1.zn; J32=dNz2.zn; J33=dNz3.zn;  J34=dNz4.zn;
   Jmat={{1,1,1,1},{J11,J12,J13,J14},
         {J21,J22,J23,J24},{J31,J32,J33,J34}};
   Jdet=(J13*J22-J12*J23+J14*J23-J14*J22+J12*J24-J13*J24)*J31-
        (J13*J21-J11*J23+J14*J23-J14*J21+J11*J24-J13*J24)*J32+
        (J12*J21-J11*J22+J14*J22-J14*J21+J11*J24-J12*J24)*J33-
        (J12*J21-J11*J22+J13*J22-J13*J21+J11*J23-J12*J23)*J34;
   Jinv={{J22*(J34-J33)-J23*(J34-J32)+J24*(J33-J32),
         -J12*(J34-J33)+J13*(J34-J32)-J14*(J33-J32),
          J12*(J24-J23)-J13*(J24-J22)+J14*(J23-J22)},
        {-J21*(J34-J33)+J23*(J34-J31)-J24*(J33-J31),
          J11*(J34-J33)-J13*(J34-J31)+J14*(J33-J31),
         -J11*(J24-J23)+J13*(J24-J21)-J14*(J23-J21)},
        { J21*(J34-J32)-J22*(J34-J31)+J24*(J32-J31),
         -J11*(J34-J32)+J12*(J34-J31)-J14*(J32-J31),
          J11*(J24-J22)-J12*(J24-J21)+J14*(J22-J21)},
        {-J21*(J33-J32)+J22*(J33-J31)-J23*(J32-J31),
          J11*(J33-J32)-J12*(J33-J31)+J13*(J32-J31),
         -J11*(J23-J22)+J12*(J23-J21)-J13*(J22-J21)}};
   {Bx,By,Bz}=Transpose[Jinv].{dNz1,dNz2,dNz3,dNz4}/Jdet;
   Return[{Bx,By,Bz,Jdet}]
   ];
```

Testing the Stiffness Module


```
ClearAll[Em,ν]; Em=96; ν=1/3;
Emat=Em/((1+ν)*(1-2*ν))*{{1-ν,ν,ν,0,0,0},
  {ν,1-ν,ν,0,0,0},{ν,ν,1-ν,0,0,0},{0,0,0,1/2-ν,0,0},
  {0,0,0,0,1/2-ν,0},{0,0,0,0,0, 1/2-ν}};
Print["Emat=",Emat//MatrixForm];
ncoor={{2,3,4},{6,3,2},{2,5,1},{4,3,6}};
Ke=IsoTetr4Stiffness[ncoor,Emat,{},{False}];
Print["Ke=",Ke//MatrixForm];
Print["eigs of Ke=",Chop[Eigenvalues[N[Ke]]]];
```

$$\text{Emat} = \begin{pmatrix} 144 & 72 & 72 & 0 & 0 & 0 \\ 72 & 144 & 72 & 0 & 0 & 0 \\ 72 & 72 & 144 & 0 & 0 & 0 \\ 0 & 0 & 0 & 36 & 0 & 0 \\ 0 & 0 & 0 & 0 & 36 & 0 \\ 0 & 0 & 0 & 0 & 0 & 36 \end{pmatrix}$$

$$\text{Ke} = \begin{pmatrix} 149 & 108 & 24 & -1 & 6 & 12 & -54 & -48 & 0 & -94 & -66 & -36 \\ 108 & 344 & 54 & -24 & 104 & 42 & -24 & -216 & -12 & -60 & -232 & -84 \\ 24 & 54 & 113 & 0 & 30 & 35 & 0 & -24 & -54 & -24 & -60 & -94 \\ -1 & -24 & 0 & 29 & -18 & -12 & -18 & 24 & 0 & -10 & 18 & 12 \\ 6 & 104 & 30 & -18 & 44 & 18 & 12 & -72 & -12 & 0 & -76 & -36 \\ 12 & 42 & 35 & -12 & 18 & 29 & 0 & -24 & -18 & 0 & -36 & -46 \\ -54 & -24 & 0 & -18 & 12 & 0 & 36 & 0 & 0 & 36 & 12 & 0 \\ -48 & -216 & -24 & 24 & -72 & -24 & 0 & 144 & 0 & 24 & 144 & 48 \\ 0 & -12 & -54 & 0 & -12 & -18 & 0 & 0 & 36 & 0 & 24 & 36 \\ -94 & -60 & -24 & -10 & 0 & 0 & 36 & 24 & 0 & 68 & 36 & 24 \\ -66 & -232 & -60 & 18 & -76 & -36 & 12 & 144 & 24 & 36 & 164 & 72 \\ -36 & -84 & -94 & 12 & -36 & -46 & 0 & 48 & 36 & 24 & 72 & 104 \end{pmatrix}$$

eigs of Ke = {777.175, 201.363, 197.273, 42.9431, 21.3643, 19.8821, 0, 0, 0, 0, 0, 0}